Axioms Systems for Category Theory in Free Logic

Christoph Benzmüller[1] and Dana S. Scott[2]

[1] *University of Luxemburg, Luxemburg & Freie Universität Berlin, Germany*
[2] *Visiting Scholar at University of Califormia, Berkeley, USA*

August 16, 2018

Contents

1 Introduction

This document provides a concise overview on the core results of our previous work [2, 3, 1] on the exploration of axiom systems for category theory. Extending the previous studies we include one further axiomatic theory in our experiments. This additional theory has been suggested by Mac Lane [5] in 1948. We show that the axioms proposed by Mac Lane are equivalent to the ones studied in [3], which includes an axioms set suggested by Scott [6] in the 1970s and another axioms set proposed by Freyd and Scedrov [4] in 1990, which we slightly modified in [3] to remedy a minor technical issue.

The explanations given below are minimal, for more details we refer to the referenced papers, in particular, to [3].

2 Embedding of Free Logic in HOL

We introduce a shallow semantical embedding of free logic [3] in Isabelle/HOL. Definite description is omitted, since it is not needed in the studies below and also since the definition provided in [1] introduces the here undesired commitment that at least one non-existing element of type i is a priori given. We here want to consider this an optional condition.

typedecl i — Type for individuals
consts *fExistence*:: $i \Rightarrow bool$ (E) — Existence/definedness predicate in free logic

abbreviation *fNot* $(\neg)$ **where** $\neg\varphi \equiv \neg\varphi$
abbreviation *fImpl* (**infixr** $\rightarrow$ *13*) **where** $\varphi \rightarrow \psi \equiv \varphi \longrightarrow \psi$
abbreviation *fId* (**infixr** $=$ *25*) **where** $l = r \equiv l = r$
abbreviation *fAll* $(\forall)$ **where** $\forall\,\Phi \equiv \forall x.\ E\ x \longrightarrow \Phi\ x$
abbreviation *fAllBi* (**binder** $\forall$ [*8*]*9*) **where** $\forall\,x.\ \varphi\ x \equiv \forall\,\varphi$
abbreviation *fOr* (**infixr** $\vee$ *21*) **where** $\varphi \vee \psi \equiv (\neg\varphi) \rightarrow \psi$
abbreviation *fAnd* (**infixr** $\wedge$ *22*) **where** $\varphi \wedge \psi \equiv \neg(\neg\varphi \vee \neg\psi)$
abbreviation *fImpli* (**infixr** $\leftarrow$ *13*) **where** $\varphi \leftarrow \psi \equiv \psi \rightarrow \varphi$
abbreviation *fEquiv* (**infixr** $\leftrightarrow$ *15*) **where** $\varphi \leftrightarrow \psi \equiv (\varphi \rightarrow \psi) \wedge (\psi \rightarrow \varphi)$
abbreviation *fEx* $(\exists)$ **where** $\exists\,\Phi \equiv \neg(\forall\,(\lambda y.\ \neg(\Phi\ y)))$
abbreviation *fExiBi* (**binder** $\exists$ [*8*]*9*) **where** $\exists\,x.\ \varphi\ x \equiv \exists\,\varphi$

3 Some Basic Notions in Category Theory

Morphisms in the category are modeled as objects of type i. We introduce three partial functions, *dom* (domain), *cod* (codomain), and morphism composition $(\cdot)$.

For composition we assume set-theoretical composition here (i.e., functional composition from right to left).

consts
domain:: $i \Rightarrow i$ (*dom* - [*108*] *109*)
codomain:: $i \Rightarrow i$ (*cod* - [*110*] *111*)
composition:: $i \Rightarrow i \Rightarrow i$ (**infix** $\cdot$ *110*)

— Kleene Equality
abbreviation *KlEq* (**infixr** $\cong$ *56*) **where** $x \cong y \equiv (E\ x \vee E\ y) \rightarrow x = y$
— Existing Identity
abbreviation *ExId* (**infixr** $\simeq$ *56*) **where** $x \simeq y \equiv (E\ x \wedge E\ y \wedge x = y)$

— Identity-morphism: see also p. 4. of [4].
abbreviation *ID* $i \equiv (\forall\,x.\ E(i\cdot x) \rightarrow i\cdot x \cong x) \wedge (\forall\,x.\ E(x\cdot i) \rightarrow x\cdot i \cong x)$
— Identity-morphism: Mac Lane's definition, the same as ID except for notion of equality.
abbreviation *IDMcL* $\varrho \equiv (\forall\,\alpha.\ E(\varrho\cdot\alpha) \rightarrow \varrho\cdot\alpha = \alpha) \wedge (\forall\,\beta.\ E(\beta\cdot\varrho) \rightarrow \beta\cdot\varrho = \beta)$

— The two notions of identity-morphisms are obviously equivalent.
lemma *IDPredicates*: *ID* $\equiv$ *IDMcL* **by** *auto*

4 The Axioms Sets studied by Benzmüller and Scott [3]

4.1 AxiomsSet1

AxiomsSet1 generalizes the notion of a monoid by introducing a partial, strict binary composition operation "·". The existence of left and right identity elements is addressed in axioms C_i and D_i. The notions of *dom* (domain) and *cod* (codomain) abstract from their common meaning in the context of sets. In category theory we work with just a single type of objects (the type i in our setting) and therefore identity morphisms are employed to suitably characterize their meanings.

locale *AxiomsSet1* =
assumes
S_i: $E(x{\cdot}y) \rightarrow (E\ x \wedge E\ y)$ **and**
E_i: $E(x{\cdot}y) \leftarrow (E\ x \wedge E\ y \wedge (\exists z.\ z{\cdot}z \cong z \wedge x{\cdot}z \cong x \wedge z{\cdot}y \cong y))$ **and**
A_i: $x{\cdot}(y{\cdot}z) \cong (x{\cdot}y){\cdot}z$ **and**
C_i: $\forall y.\exists i.\ ID\ i \wedge i{\cdot}y \cong y$ **and**
D_i: $\forall x.\exists j.\ ID\ j \wedge x{\cdot}j \cong x$
begin
lemma *True* **nitpick** [*satisfy*] **oops** — Consistency
lemma assumes $\exists x.\ \neg(E\ x)$ **shows** *True* **nitpick** [*satisfy*] **oops** — Consistency
lemma assumes $(\exists x.\ \neg(E\ x)) \wedge (\exists x.\ (E\ x))$ **shows** *True* **nitpick** [*satisfy*] **oops** — Consistency

lemma $E_i Impl$: $E(x{\cdot}y) \rightarrow (E\ x \wedge E\ y \wedge (\exists z.\ z{\cdot}z \cong z \wedge x{\cdot}z \cong x \wedge z{\cdot}y \cong y))$ **by** (*metis* A_i C_i S_i)
— Uniqueness of i and j in the latter two axioms.
lemma UC_i: $\forall y.\exists i.\ ID\ i \wedge i{\cdot}y \cong y \wedge (\forall j.(ID\ j \wedge j{\cdot}y \cong y) \rightarrow i \cong j)$ **by** (*smt* A_i C_i S_i)
lemma UD_i: $\forall x.\exists j.\ ID\ j \wedge x{\cdot}j \cong x \wedge (\forall i.(ID\ i \wedge x{\cdot}i \cong x) \rightarrow j \cong i)$ **by** (*smt* A_i D_i S_i)
— But i and j need not to equal.
lemma $(\exists C\ D.\ (\forall y.\ ID\ (C\ y) \wedge (C\ y){\cdot}y \cong y) \wedge (\forall x.\ ID\ (D\ x) \wedge x{\cdot}(D\ x) \cong x) \wedge \neg(D = C))$
nitpick [*satisfy*] **oops** — Model found
lemma $(\exists x.\ E\ x) \wedge (\exists C\ D.\ (\forall y.\ ID(C\ y) \wedge (C\ y){\cdot}y \cong y) \wedge (\forall x.\ ID(D\ x) \wedge x{\cdot}(D\ x) \cong x) \wedge \neg(D = C))$
nitpick [*satisfy*] **oops** — Model found
end

4.2 AxiomsSet2

AxiomsSet2 is developed from AxiomsSet1 by Skolemization of the existentially quantified variables i and j in axioms C_i and D_i. We can argue semantically that every model of AxiomsSet1 has such functions. Hence, we get a conservative extension of AxiomsSet1. The strictness axiom S is extended, so that strictness is now also postulated for the new Skolem functions *dom* and *cod*.

locale *AxiomsSet2* =
assumes
S_{ii}: $(E(x{\cdot}y) \rightarrow (E\ x \wedge E\ y)) \wedge (E(dom\ x) \rightarrow E\ x) \wedge (E(cod\ y) \rightarrow E\ y)$ **and**
E_{ii}: $E(x{\cdot}y) \leftarrow (E\ x \wedge E\ y \wedge (\exists z.\ z{\cdot}z \cong z \wedge x{\cdot}z \cong x \wedge z{\cdot}y \cong y))$ **and**
A_{ii}: $x{\cdot}(y{\cdot}z) \cong (x{\cdot}y){\cdot}z$ **and**
C_{ii}: $E\ y \rightarrow (ID(cod\ y) \wedge (cod\ y){\cdot}y \cong y)$ **and**
D_{ii}: $E\ x \rightarrow (ID(dom\ x) \wedge x{\cdot}(dom\ x) \cong x)$
begin
lemma *True* **nitpick** [*satisfy*] **oops** — Consistency
lemma assumes $\exists x.\ \neg(E\ x)$ **shows** *True* **nitpick** [*satisfy*] **oops** — Consistency
lemma assumes $(\exists x.\ \neg(E\ x)) \wedge (\exists x.\ (E\ x))$ **shows** *True* **nitpick** [*satisfy*] **oops** — Consistency

lemma $E_{ii} Impl$: $E(x{\cdot}y) \rightarrow (E\ x \wedge E\ y \wedge (\exists z.\ z{\cdot}z \cong z \wedge x{\cdot}z \cong x \wedge z{\cdot}y \cong y))$ **by** (*metis* A_{ii} C_{ii} S_{ii})
lemma *domTotal*: $E\ x \rightarrow E(dom\ x)$ **by** (*metis* D_{ii} S_{ii})
lemma *codTotal*: $E\ x \rightarrow E(cod\ x)$ **by** (*metis* C_{ii} S_{ii})
end

4.2.1 AxiomsSet2 entails AxiomsSet1

context *AxiomsSet2*
begin

lemma S_i: $E(x{\cdot}y) \rightarrow (E\ x \wedge E\ y)$ **using** S_{ii} **by** *blast*
lemma E_i: $E(x{\cdot}y) \leftarrow (E\ x \wedge E\ y \wedge (\exists z.\ z{\cdot}z \cong z \wedge x{\cdot}z \cong x \wedge z{\cdot}y \cong y))$ **using** E_{ii} **by** *blast*
lemma A_i: $x{\cdot}(y{\cdot}z) \cong (x{\cdot}y){\cdot}z$ **using** A_{ii} **by** *blast*
lemma C_i: $\forall y.\exists i.\ ID\ i \wedge i{\cdot}y \cong y$ **by** $(metis\ C_{ii}\ S_{ii})$
lemma D_i: $\forall x.\exists j.\ ID\ j \wedge x{\cdot}j \cong x$ **by** $(metis\ D_{ii}\ S_{ii})$
end

4.2.2 AxiomsSet1 entails AxiomsSet2 (by semantic means)

By semantic means (Skolemization).

4.3 AxiomsSet3

In AxiomsSet3 the existence axiom E_{ii} from AxiomsSet2 is simplified by taking advantage of the two new Skolem functions *dom* and *cod*.

The left-to-right direction of existence axiom E_{iii} is implied.

locale *AxiomsSet3* =
assumes
S_{iii}: $(E(x{\cdot}y) \rightarrow (E\ x \wedge E\ y)) \wedge (E(dom\ x\ ) \rightarrow E\ x) \wedge (E(cod\ y) \rightarrow E\ y)$ **and**
E_{iii}: $E(x{\cdot}y) \leftarrow (dom\ x \cong cod\ y \wedge E(cod\ y))$ **and**
A_{iii}: $x{\cdot}(y{\cdot}z) \cong (x{\cdot}y){\cdot}z$ **and**
C_{iii}: $E\ y \rightarrow (ID(cod\ y) \wedge (cod\ y){\cdot}y \cong y)$ **and**
D_{iii}: $E\ x \rightarrow (ID(dom\ x) \wedge x{\cdot}(dom\ x) \cong x)$
begin
lemma *True* **nitpick** [*satisfy*] **oops** — Consistency
lemma assumes $\exists x.\ \neg(E\ x)$ **shows** *True* **nitpick** [*satisfy*] **oops** — Consistency
lemma assumes $(\exists x.\ \neg(E\ x)) \wedge (\exists x.\ (E\ x))$ **shows** *True* **nitpick** [*satisfy*] **oops** — Consistency

lemma $E_{iii}Impl$: $E(x{\cdot}y) \rightarrow (dom\ x \cong cod\ y \wedge E(cod\ y))$ **by** $(metis\ (full\text{-}types)\ A_{iii}\ C_{iii}\ D_{iii}\ S_{iii})$
end

4.3.1 AxiomsSet3 entails AxiomsSet2

context *AxiomsSet3*
begin
lemma S_{ii}: $(E(x{\cdot}y) \rightarrow (E\ x \wedge E\ y)) \wedge (E(dom\ x\ ) \rightarrow E\ x) \wedge (E(cod\ y) \rightarrow E\ y)$ **using** S_{iii} **by** *blast*
lemma E_{ii}: $E(x{\cdot}y) \leftarrow (E\ x \wedge E\ y \wedge (\exists z.\ z{\cdot}z \cong z \wedge x{\cdot}z \cong x \wedge z{\cdot}y \cong y))$ **by** $(metis\ A_{iii}\ C_{iii}\ D_{iii}\ E_{iii}\ S_{iii})$
lemma A_{ii}: $x{\cdot}(y{\cdot}z) \cong (x{\cdot}y){\cdot}z$ **using** A_{iii} **by** *blast*
lemma C_{ii}: $E\ y \rightarrow (ID(cod\ y) \wedge (cod\ y){\cdot}y \cong y)$ **using** C_{iii} **by** *auto*
lemma D_{ii}: $E\ x \rightarrow (ID(dom\ x) \wedge x{\cdot}(dom\ x) \cong x)$ **using** D_{iii} **by** *auto*
end

4.3.2 AxiomsSet2 entails AxiomsSet3

context *AxiomsSet2*
begin
lemma S_{iii}: $(E(x{\cdot}y) \rightarrow (E\ x \wedge E\ y)) \wedge (E(dom\ x) \rightarrow E\ x) \wedge (E(cod\ y) \rightarrow E\ y)$ **using** S_{ii} **by** *blast*
lemma E_{iii}: $E(x{\cdot}y) \leftarrow (dom\ x \cong cod\ y \wedge E(cod\ y))$ **by** $(metis\ C_{ii}\ D_{ii}\ E_{ii}\ S_{ii})$
lemma A_{iii}: $x{\cdot}(y{\cdot}z) \cong (x{\cdot}y){\cdot}z$ **using** A_{ii} **by** *blast*
lemma C_{iii}: $E\ y \rightarrow (ID(cod\ y) \wedge (cod\ y){\cdot}y \cong y)$ **using** C_{ii} **by** *auto*
lemma D_{iii}: $E\ x \rightarrow (ID(dom\ x) \wedge x{\cdot}(dom\ x) \cong x)$ **using** D_{ii} **by** *auto*
end

4.4 The Axioms Set AxiomsSet4

AxiomsSet4 simplifies the axioms C_{iii} and D_{iii}. However, as it turned out, these simplifications also require the existence axiom E_{iii} to be strengthened into an equivalence.

locale *AxiomsSet4* =

assumes
S_{iv}: $(E(x{\cdot}y) \rightarrow (E\ x \wedge E\ y)) \wedge (E(dom\ x) \rightarrow E\ x) \wedge (E(cod\ y) \rightarrow E\ y)$ **and**
E_{iv}: $E(x{\cdot}y) \leftrightarrow (dom\ x \cong cod\ y \wedge E(cod\ y))$ **and**
A_{iv}: $x{\cdot}(y{\cdot}z) \cong (x{\cdot}y){\cdot}z$ **and**
C_{iv}: $(cod\ y){\cdot}y \cong y$ **and**
D_{iv}: $x{\cdot}(dom\ x) \cong x$
begin
lemma *True* **nitpick** [*satisfy*] **oops** — Consistency
lemma assumes $\exists x.\ \neg(E\ x)$ **shows** *True* **nitpick** [*satisfy*] **oops** — Consistency
lemma assumes $(\exists x.\ \neg(E\ x)) \wedge (\exists x.\ (E\ x))$ **shows** *True* **nitpick** [*satisfy*] **oops** — Consistency
end

4.4.1 AxiomsSet4 entails AxiomsSet3

context *AxiomsSet4*
begin
lemma S_{iii}: $(E(x{\cdot}y) \rightarrow (E\ x \wedge E\ y)) \wedge (E(dom\ x) \rightarrow E\ x) \wedge (E(cod\ y) \rightarrow E\ y)$ **using** S_{iv} **by** *blast*
lemma E_{iii}: $E(x{\cdot}y) \leftarrow (dom\ x \cong cod\ y \wedge (E(cod\ y)))$ **using** E_{iv} **by** *blast*
lemma A_{iii}: $x{\cdot}(y{\cdot}z) \cong (x{\cdot}y){\cdot}z$ **using** A_{iv} **by** *blast*
lemma C_{iii}: $E\ y \rightarrow (ID(cod\ y) \wedge (cod\ y){\cdot}y \cong y)$ **by** $(metis\ C_{iv}\ D_{iv}\ E_{iv})$
lemma D_{iii}: $E\ x \rightarrow (ID(dom\ x) \wedge x{\cdot}(dom\ x) \cong x)$ **by** $(metis\ C_{iv}\ D_{iv}\ E_{iv})$
end

4.4.2 AxiomsSet3 entails AxiomsSet4

context *AxiomsSet3*
begin
lemma S_{iv}: $(E(x{\cdot}y) \rightarrow (E\ x \wedge E\ y)) \wedge (E(dom\ x\ ) \rightarrow E\ x) \wedge (E(cod\ y) \rightarrow E\ y)$ **using** S_{iii} **by** *blast*
lemma E_{iv}: $E(x{\cdot}y) \leftrightarrow (dom\ x \cong cod\ y \wedge E(cod\ y))$ **by** $(metis\ (full\text{-}types)\ A_{iii}\ C_{iii}\ D_{iii}\ E_{iii}\ S_{iii})$
lemma A_{iv}: $x{\cdot}(y{\cdot}z) \cong (x{\cdot}y){\cdot}z$ **using** A_{iii} **by** *blast*
lemma C_{iv}: $(cod\ y){\cdot}y \cong y$ **using** $C_{iii}\ S_{iii}$ **by** *blast*
lemma D_{iv}: $x{\cdot}(dom\ x) \cong x$ **using** $D_{iii}\ S_{iii}$ **by** *blast*
end

4.5 AxiomsSet5

AxiomsSet5 has been proposed by Scott [6] in the 1970s. This set of axioms is equivalent to the axioms set presented by Freyd and Scedrov in their textbook "Categories, Allegories" [4] when encoded in free logic, corrected/adapted and further simplified, see Section 5.

locale *AxiomsSet5* =
assumes
S1: $E(dom\ x) \rightarrow E\ x$ **and**
S2: $E(cod\ y) \rightarrow E\ y$ **and**
S3: $E(x{\cdot}y) \leftrightarrow dom\ x \simeq cod\ y$ **and**
S4: $x{\cdot}(y{\cdot}z) \cong (x{\cdot}y){\cdot}z$ **and**
S5: $(cod\ y){\cdot}y \cong y$ **and**
S6: $x{\cdot}(dom\ x) \cong x$
begin
lemma *True* **nitpick** [*satisfy*] **oops** — Consistency
lemma assumes $\exists x.\ \neg(E\ x)$ **shows** *True* **nitpick** [*satisfy*] **oops** — Consistency
lemma assumes $(\exists x.\ \neg(E\ x)) \wedge (\exists x.\ (E\ x))$ **shows** *True* **nitpick** [*satisfy*] **oops** — Consistency
end

4.5.1 AxiomsSet5 entails AxiomsSet4

context *AxiomsSet5*
begin
lemma S_{iv}: $(E(x{\cdot}y) \rightarrow (E\ x \wedge E\ y)) \wedge (E(dom\ x\ ) \rightarrow E\ x) \wedge (E(cod\ y) \rightarrow E\ y)$ **using** *S1 S2 S3* **by** *blast*
lemma E_{iv}: $E(x{\cdot}y) \leftrightarrow (dom\ x \cong cod\ y \wedge E(cod\ y))$ **using** *S3* **by** *metis*
lemma A_{iv}: $x{\cdot}(y{\cdot}z) \cong (x{\cdot}y){\cdot}z$ **using** *S4* **by** *blast*

lemma C_{iv}: $(cod\ y){\cdot}y \cong y$ **using** *S5* **by** *blast*
lemma D_{iv}: $x{\cdot}(dom\ x) \cong x$ **using** *S6* **by** *blast*
end

4.5.2 AxiomsSet4 entails AxiomsSet5

context *AxiomsSet4*
begin
lemma *S1*: $E(dom\ x) \rightarrow E\ x$ **using** S_{iv} **by** *blast*
lemma *S2*: $E(cod\ y) \rightarrow E\ y$ **using** S_{iv} **by** *blast*
lemma *S3*: $E(x{\cdot}y) \leftrightarrow dom\ x \simeq cod\ y$ **using** E_{iv} **by** *metis*
lemma *S4*: $x{\cdot}(y{\cdot}z) \cong (x{\cdot}y){\cdot}z$ **using** A_{iv} **by** *blast*
lemma *S5*: $(cod\ y){\cdot}y \cong y$ **using** C_{iv} **by** *blast*
lemma *S6*: $x{\cdot}(dom\ x) \cong x$ **using** D_{iv} **by** *blast*
end

5 The Axioms Sets by Freyd and Scedrov [4]

5.1 AxiomsSet6

The axioms by Freyd and Scedrov [4] in our notation, when being corrected (cf. the modification in axiom A1).

Freyd and Scedrov employ a different notation for *dom x* and *cod x*. They denote these operations by $\Box x$ and $x\Box$. Moreover, they employ diagrammatic composition instead of the set-theoretic definition (functional composition from right to left) used so far. We leave it to the reader to verify that their axioms corresponds to the axioms presented here modulo an appropriate conversion of notation.

locale *AxiomsSet6* =
assumes
A1: $E(x{\cdot}y) \leftrightarrow dom\ x \simeq cod\ y$ **and**
A2a: $cod(dom\ x) \cong dom\ x$ **and**
A2b: $dom(cod\ y) \cong cod\ y$ **and**
A3a: $x{\cdot}(dom\ x) \cong x$ **and**
A3b: $(cod\ y){\cdot}y \cong y$ **and**
A4a: $dom(x{\cdot}y) \cong dom((dom\ x){\cdot}y)$ **and**
A4b: $cod(x{\cdot}y) \cong cod(x{\cdot}(cod\ y))$ **and**
A5: $x{\cdot}(y{\cdot}z) \cong (x{\cdot}y){\cdot}z$
begin
lemma *True* **nitpick** [*satisfy*] **oops** — Consistency
lemma assumes $\exists x.\ \neg(E\ x)$ **shows** *True* **nitpick** [*satisfy*] **oops** — Consistency
lemma assumes $(\exists x.\ \neg(E\ x)) \wedge (\exists x.\ (E\ x))$ **shows** *True* **nitpick** [*satisfy*] **oops** — Consistency
end

5.1.1 AxiomsSet6 entails AxiomsSet5

context *AxiomsSet6*
begin
lemma *S1*: $E(dom\ x) \rightarrow E\ x$ **by** (*metis A1 A2a A3a*)
lemma *S2*: $E(cod\ y) \rightarrow E\ y$ **using** *A1 A2b A3b* **by** *metis*
lemma *S3*: $E(x{\cdot}y) \leftrightarrow dom\ x \simeq cod\ y$ **by** (*metis A1*)
lemma *S4*: $x{\cdot}(y{\cdot}z) \cong (x{\cdot}y){\cdot}z$ **using** *A5* **by** *blast*
lemma *S5*: $(cod\ y){\cdot}y \cong y$ **using** *A3b* **by** *blast*
lemma *S6*: $x{\cdot}(dom\ x) \cong x$ **using** *A3a* **by** *blast*

lemma *A4aRedundant*: $dom(x{\cdot}y) \cong dom((dom\ x){\cdot}y)$ **using** *A1 A2a A3a A5* **by** *metis*
lemma *A4bRedundant*: $cod(x{\cdot}y) \cong cod(x{\cdot}(cod\ y))$ **using** *A1 A2b A3b A5* **by** *smt*
lemma *A2aRedundant*: $cod(dom\ x) \cong dom\ x$ **using** *A1 A3a A3b A4a A4b* **by** *smt*
lemma *A2bRedundant*: $dom(cod\ y) \cong cod\ y$ **using** *A1 A3a A3b A4a A4b* **by** *smt*
end

5.1.2 AxiomsSet5 entails AxiomsSet6

context *AxiomsSet5*
begin
lemma *A1*: $E(x{\cdot}y) \leftrightarrow dom\ x \simeq cod\ y$ **using** *S3* **by** *blast*
lemma *A2*: $cod(dom\ x) \cong dom\ x$ **by** (*metis S1 S2 S3 S6*)
lemma *A2b*: $dom(cod\ y) \cong cod\ y$ **using** *S1 S2 S3 S5* **by** *metis*
lemma *A3a*: $x{\cdot}(dom\ x) \cong x$ **using** *S6* **by** *auto*
lemma *A3b*: $(cod\ y){\cdot}y \cong y$ **using** *S5* **by** *blast*
lemma *A4a*: $dom(x{\cdot}y) \cong dom((dom\ x){\cdot}y)$ **by** (*metis S1 S3 S4 S5 S6*)
lemma *A4b*: $cod(x{\cdot}y) \cong cod(x{\cdot}(cod\ y))$ **by** (*metis* (*full-types*) *S2 S3 S4 S5 S6*)
lemma *A5*: $x{\cdot}(y{\cdot}z) \cong (x{\cdot}y){\cdot}z$ **using** *S4* **by** *blast*
end

5.2 AxiomsSet7 (technically flawed)

The axioms by Freyd and Scedrov in our notation, without the suggested correction of axiom A1. This axioms set is technically flawed when encoded in our given context. It leads to a constricted inconsistency.

locale *AxiomsSet7* =
assumes
A1: $E(x{\cdot}y) \leftrightarrow dom\ x \cong cod\ y$ **and**
A2a: $cod(dom\ x) \cong dom\ x$ **and**
A2b: $dom(cod\ y) \cong cod\ y$ **and**
A3a: $x{\cdot}(dom\ x) \cong x$ **and**
A3b: $(cod\ y){\cdot}y \cong y$ **and**
A4a: $dom(x{\cdot}y) \cong dom((dom\ x){\cdot}y)$ **and**
A4b: $cod(x{\cdot}y) \cong cod(x{\cdot}(cod\ y))$ **and**
A5: $x{\cdot}(y{\cdot}z) \cong (x{\cdot}y){\cdot}z$
begin
lemma *True* **nitpick** [*satisfy*] **oops** — Consistency

lemma *InconsistencyAutomatic*: $(\exists x.\ \neg(E\ x)) \rightarrow False$ **by** (*metis A1 A2a A3a*) — Inconsistency
lemma $\forall x.\ E\ x$ **using** *InconsistencyAutomatic* **by** *auto*

lemma *InconsistencyInteractive*:
assumes *NEx*: $\exists x.\ \neg(E\ x)$ **shows** *False*
proof −
obtain *a* **where** *1*: $\neg(E\ a)$ **using** *NEx* **by** *auto*
have *2*: $a{\cdot}(dom\ a) \cong a$ **using** *A3a* **by** *blast*
have *3*: $\neg(E(a{\cdot}(dom\ a)))$ **using** *1 2* **by** *metis*
have *4*: $E(a{\cdot}(dom\ a)) \leftrightarrow dom\ a \cong cod(dom\ a)$ **using** *A1* **by** *blast*
have *5*: $cod(dom\ a) \cong dom\ a$ **using** *A2a* **by** *blast*
have *6*: $E(a{\cdot}(dom\ a)) \leftrightarrow dom\ a \cong dom\ a$ **using** *4 5* **by** *auto*
have *7*: $E(a{\cdot}(dom\ a))$ **using** *6* **by** *blast*
then show *?thesis* **using** *7 3* **by** *blast*
qed
end

5.3 AxiomsSet7orig (technically flawed)

The axioms by Freyd and Scedrov in their original notation, without the suggested correction of axiom A1.
We present the constricted inconsistency argument from above once again, but this time in the original notation of Freyd and Scedrov.

locale *AxiomsSet7orig* =
fixes
source:: $i \Rightarrow i$ (□- [*108*] *109*) **and**
target:: $i \Rightarrow i$ (-□ [*110*] *111*) **and**

compositionF:: $i \Rightarrow i \Rightarrow i$ (**infix** $\cdot$ *110*)
assumes
A1: $E(x \cdot y) \leftrightarrow (x\Box \cong \Box y)$ **and**
A2a: $((\Box x)\Box) \cong \Box x$ **and**
A2b: $\Box(x\Box) \cong \Box x$ **and**
A3a: $(\Box x) \cdot x \cong x$ **and**
A3b: $x \cdot (x\Box) \cong x$ **and**
A4a: $\Box(x \cdot y) \cong \Box(x \cdot (\Box y))$ **and**
A4b: $(x \cdot y)\Box \cong ((x\Box) \cdot y)\Box$ **and**
A5: $x \cdot (y \cdot z) \cong (x \cdot y) \cdot z$
begin
lemma *True* **nitpick** [*satisfy*] **oops** — Consistency

lemma *InconsistencyAutomatic*: $(\exists x. \neg(E\ x)) \rightarrow$ *False* **by** (*metis A1 A2a A3a*) — Inconsistency
lemma $\forall x.\ E\ x$ **using** *InconsistencyAutomatic* **by** *auto*

lemma *InconsistencyInteractive*:
assumes *NEx*: $\exists x. \neg(E\ x)$ **shows** *False*
proof −
obtain *a* **where** *1*: $\neg(E\ a)$ **using** *assms* **by** *auto*
have *2*: $(\Box a) \cdot a \cong a$ **using** *A3a* **by** *blast*
have *3*: $\neg(E((\Box a) \cdot a))$ **using** *1 2* **by** *metis*
have *4*: $E((\Box a) \cdot a) \leftrightarrow (\Box a)\Box \cong \Box a$ **using** *A1* **by** *blast*
have *5*: $(\Box a)\Box \cong \Box a$ **using** *A2a* **by** *blast*
have *6*: $E((\Box a) \cdot a)$ **using** *4 5* **by** *blast*
then show *?thesis* **using** *6 3* **by** *blast*
qed
end

5.4 AxiomsSet8 (algebraic reading, still technically flawed)

The axioms by Freyd and Scedrov in our notation again, but this time we adopt an algebraic reading of the free variables, meaning that they range over existing morphisms only.

locale *AxiomsSet8* =
assumes
B1: $\forall x. \forall y.\ E(x \cdot y) \leftrightarrow dom\ x \cong cod\ y$ **and**
B2a: $\forall x.\ cod(dom\ x) \cong dom\ x$ **and**
B2b: $\forall y.\ dom(cod\ y) \cong cod\ y$ **and**
B3a: $\forall x.\ x \cdot (dom\ x) \cong x$ **and**
B3b: $\forall y.\ (cod\ y) \cdot y \cong y$ **and**
B4a: $\forall x. \forall y.\ dom(x \cdot y) \cong dom((dom\ x) \cdot y)$ **and**
B4b: $\forall x. \forall y.\ cod(x \cdot y) \cong cod(x \cdot (cod\ y))$ **and**
B5: $\forall x. \forall y. \forall z.\ x \cdot (y \cdot z) \cong (x \cdot y) \cdot z$
begin
lemma *True* **nitpick** [*satisfy*] **oops** — Consistency
lemma assumes $\exists x. \neg(E\ x)$ **shows** *True* **nitpick** [*satisfy*] **oops** — Consistency
lemma assumes $(\exists x. \neg(E\ x)) \wedge (\exists x. (E\ x))$ **shows** *True* **nitpick** [*satisfy*] **oops** — Consistency
end

None of the axioms in AxiomsSet5 are implied.

context *AxiomsSet8*
begin
lemma *S1*: $E(dom\ x) \rightarrow E\ x$ **nitpick oops** — Nitpick finds a countermodel
lemma *S2*: $E(cod\ y) \rightarrow E\ y$ **nitpick oops** — Nitpick finds a countermodel
lemma *S3*: $E(x \cdot y) \leftrightarrow dom\ x \simeq cod\ y$ **nitpick oops** — Nitpick finds a countermodel
lemma *S4*: $x \cdot (y \cdot z) \cong (x \cdot y) \cdot z$ **nitpick oops** — Nitpick finds a countermodel
lemma *S5*: $(cod\ y) \cdot y \cong y$ **nitpick oops** — Nitpick finds a countermodel
lemma *S6*: $x \cdot (dom\ x) \cong x$ **nitpick oops** — Nitpick finds a countermodel
end

5.5 AxiomsSet8Strict (algebraic reading)

The situation changes when strictness conditions are postulated. Note that in the algebraic framework of Freyd and Scedrov such conditions have to be assumed as given in the logic, while here we can explicitly encode them as axioms.

locale *AxiomsSet8Strict* = *AxiomsSet8* +
assumes
B0a: $E(x{\cdot}y) \rightarrow (E\ x \wedge E\ y)$ **and**
B0b: $E(dom\ x) \rightarrow E\ x$ **and**
B0c: $E(cod\ x) \rightarrow E\ x$
begin
lemma *True* **nitpick** [*satisfy*] **oops** — Consistency
lemma assumes $\exists x.\ \neg(E\ x)$ **shows** *True* **nitpick** [*satisfy*] **oops** — Consistency
lemma assumes $(\exists x.\ \neg(E\ x)) \wedge (\exists x.\ (E\ x))$ **shows** *True* **nitpick** [*satisfy*] **oops** — Consistency
end

5.5.1 AxiomsSet8Strict entails AxiomsSet5

context *AxiomsSet8Strict*
begin
lemma *S1*: $E(dom\ x) \rightarrow E\ x$ **using** *B0b* **by** *blast*
lemma *S2*: $E(cod\ y) \rightarrow E\ y$ **using** *B0c* **by** *blast*
lemma *S3*: $E(x{\cdot}y) \leftrightarrow dom\ x \simeq cod\ y$ **by** (*metis B0a B0b B0c B1 B3a*)
lemma *S4*: $x{\cdot}(y{\cdot}z) \cong (x{\cdot}y){\cdot}z$ **by** (*meson B0a B5*)
lemma *S5*: $(cod\ y){\cdot}y \cong y$ **using** *B0a B3b* **by** *blast*
lemma *S6*: $x{\cdot}(dom\ x) \cong x$ **using** *B0a B3a* **by** *blast*
end

5.5.2 AxiomsSet5 entails AxiomsSet8Strict

context *AxiomsSet5*
begin
lemma *B0a*: $E(x{\cdot}y) \rightarrow (E\ x \wedge E\ y)$ **using** *S1 S2 S3* **by** *blast*
lemma *B0b*: $E(dom\ x) \rightarrow E\ x$ **using** *S1* **by** *blast*
lemma *B0c*: $E(cod\ x) \rightarrow E\ x$ **using** *S2* **by** *blast*
lemma *B1*: $\forall x.\forall y.\ E(x{\cdot}y) \leftrightarrow dom\ x \cong cod\ y$ **by** (*metis S3 S5*)
lemma *B2a*: $\forall x.\ cod(dom\ x) \cong dom\ x$ **using** *A2* **by** *blast*
lemma *B2b*: $\forall y.\ dom(cod\ y) \cong cod\ y$ **using** *A2b* **by** *blast*
lemma *B3a*: $\forall x.\ x{\cdot}(dom\ x) \cong x$ **using** *S6* **by** *blast*
lemma *B3b*: $\forall y.\ (cod\ y){\cdot}y \cong y$ **using** *S5* **by** *blast*
lemma *B4a*: $\forall x.\forall y.\ dom(x{\cdot}y) \cong dom((dom\ x){\cdot}y)$ **by** (*metis S1 S3 S4 S6*)
lemma *B4b*: $\forall x.\forall y.\ cod(x{\cdot}y) \cong cod(x{\cdot}(cod\ y))$ **by** (*metis S1 S2 S3 S4 S5*)
lemma *B5*: $\forall x.\forall y.\forall z.\ x{\cdot}(y{\cdot}z) \cong (x{\cdot}y){\cdot}z$ **using** *S4* **by** *blast*
end

5.5.3 AxiomsSet8Strict is Redundant

AxiomsSet8Strict is redundant: either the B2-axioms can be omitted or the B4-axioms.

context *AxiomsSet8Strict*
begin
lemma *B2aRedundant*: $\forall x.\ cod(dom\ x) \cong dom\ x$ **by** (*metis B0a B1 B3a*)
lemma *B2bRedundant*: $\forall y.\ dom(cod\ y) \cong cod\ y$ **by** (*metis B0a B1 B3b*)
lemma *B4aRedundant*: $\forall x.\forall y.\ dom(x{\cdot}y) \cong dom((dom\ x){\cdot}y)$ **by** (*metis B0a B0b B1 B3a B5*)
lemma *B4bRedundant*: $\forall x.\forall y.\ cod(x{\cdot}y) \cong cod(x{\cdot}(cod\ y))$ **by** (*metis B0a B0c B1 B3b B5*)
end

6 The Axioms Sets of Mac Lane [5]

We analyse the axioms set suggested by Mac Lane [5] already in 1948. As for the theory by Freyd and Scedrov above, which was developed much later, we need to assume strictness of composition to show equivalence to our previous axiom sets. Note that his complicated conditions on existence of compositions proved to be unnecessary, as we show. It shows it is hard to think about partial operations.

locale *AxiomsSetMcL* =
assumes
$C_0 : E(x\cdot y) \rightarrow (E\ x \wedge E\ y)$ **and**
$C_1 : \forall \gamma\ \beta\ \alpha.\ (E(\gamma\cdot\beta) \wedge E((\gamma\cdot\beta)\cdot\alpha)) \rightarrow E(\beta\cdot\alpha)$ **and**
$C_1{}'\!: \forall \gamma\ \beta\ \alpha.\ (E(\beta\cdot\alpha) \wedge E(\gamma\cdot(\beta\cdot\alpha))) \rightarrow E(\gamma\cdot\beta)$ **and**
$C_2 : \forall \gamma\ \beta\ \alpha.\ (E(\gamma\cdot\beta) \wedge E(\beta\cdot\alpha)) \rightarrow (E((\gamma\cdot\beta)\cdot\alpha) \wedge E(\gamma\cdot(\beta\cdot\alpha)) \wedge ((\gamma\cdot\beta)\cdot\alpha) = (\gamma\cdot(\beta\cdot\alpha)))$ **and**
$C_3 : \forall \gamma.\ \exists\, eD.\ IDMcL(eD) \wedge E(\gamma\cdot eD)$ **and**
$C_4 : \forall \gamma.\ \exists\, eR.\ IDMcL(eR) \wedge E(eR\cdot\gamma)$
begin
lemma *True* **nitpick** [*satisfy*] **oops** — Consistency
lemma assumes $\exists\, x.\ \neg(E\ x)$ **shows** *True* **nitpick** [*satisfy*] **oops** — Consistency
lemma assumes $(\exists\, x.\ \neg(E\ x)) \wedge (\exists\, x.\ (E\ x))$ **shows** *True* **nitpick** [*satisfy*] **oops** — Consistency
end

Remember that IDMcL was defined on p. 2 and proved equivalent to ID.

6.1 AxiomsSetMcL entails AxiomsSet1

context *AxiomsSetMcL*
begin
lemma S_i: $E(x\cdot y) \rightarrow (E\ x \wedge E\ y)$ **using** C_0 **by** *blast*
lemma E_i: $E(x\cdot y) \leftarrow (E\ x \wedge E\ y \wedge (\exists\, z.\ z\cdot z \cong z \wedge x\cdot z \cong x \wedge z\cdot y \cong y))$ **by** (*metis* C_2)
lemma A_i: $x\cdot(y\cdot z) \cong (x\cdot y)\cdot z$ **by** (*metis* C_1 $C_1{}'$ C_2 C_0)
lemma C_i: $\forall\, y.\exists\, i.\ ID\ i \wedge i\cdot y \cong y$ **using** C_4 **by** *fastforce*
lemma D_i: $\forall\, x.\exists\, j.\ ID\ j \wedge x\cdot j \cong x$ **using** C_3 **by** *fastforce*
end

6.2 AxiomsSet1 entails AxiomsSetMcL

context *AxiomsSet1*
begin
lemma $C_0 : E(x\cdot y) \rightarrow (E\ x \wedge E\ y)$ **using** S_i **by** *blast*
lemma $C_1 : \forall \gamma\ \beta\ \alpha.\ (E(\gamma\cdot\beta) \wedge E((\gamma\cdot\beta)\cdot\alpha)) \rightarrow E(\beta\cdot\alpha)$ **by** (*metis* A_i S_i)
lemma $C_1{}'\!: \forall \gamma\ \beta\ \alpha.\ (E(\beta\cdot\alpha) \wedge E(\gamma\cdot(\beta\cdot\alpha))) \rightarrow E(\gamma\cdot\beta)$ **by** (*metis* A_i S_i)
lemma $C_2 : \forall \gamma\ \beta\ \alpha.\ (E(\gamma\cdot\beta) \wedge E(\beta\cdot\alpha)) \rightarrow (E((\gamma\cdot\beta)\cdot\alpha) \wedge E(\gamma\cdot(\beta\cdot\alpha)) \wedge ((\gamma\cdot\beta)\cdot\alpha) = (\gamma\cdot(\beta\cdot\alpha)))$ **by** (*smt* A_i C_i E_i S_i)
lemma $C_3 : \forall \gamma.\ \exists\, eD.\ IDMcL(eD) \wedge E(\gamma\cdot eD)$ **using** D_i **by** *force*
lemma $C_4 : \forall \gamma.\ \exists\, eR.\ IDMcL(eR) \wedge E(eR\cdot\gamma)$ **using** C_i **by** *force*
end

6.3 Skolemization of the Axioms of Mac Lane

Mac Lane employs diagrammatic composition instead of the set-theoretic definition as used in our axiom sets. As we have seen above, this is not a problem as long as composition is the only primitive. But when adding the Skolem terms *dom* and *cod* care must be taken and we should actually transform all axioms into a common form. Below we address this (in a minimal way) by using *dom* in axiom C_3s and *cod* in axiom C_4s, which is opposite of what Mac Lane proposed. For this axioms set we then show equivalence to AxiomsSet1/2/5.

locale *SkolemizedAxiomsSetMcL* =
assumes
$C_0s : (E(x\cdot y) \rightarrow (E\ x \wedge E\ y)) \wedge (E(dom\ x) \rightarrow E\ x) \wedge (E(cod\ y) \rightarrow E\ y)$ **and**

$C_1 s : \forall \gamma\ \beta\ \alpha.\ (E(\gamma \cdot \beta) \wedge E((\gamma \cdot \beta) \cdot \alpha)) \rightarrow E(\beta \cdot \alpha)$ **and**
$C_1{}'s: \forall \gamma\ \beta\ \alpha.\ (E(\beta \cdot \alpha) \wedge E(\gamma \cdot (\beta \cdot \alpha))) \rightarrow E(\gamma \cdot \beta)$ **and**
$C_2 s : \forall \gamma\ \beta\ \alpha.\ (E(\gamma \cdot \beta) \wedge E(\beta \cdot \alpha)) \rightarrow (E((\gamma \cdot \beta) \cdot \alpha) \wedge E(\gamma \cdot (\beta \cdot \alpha)) \wedge ((\gamma \cdot \beta) \cdot \alpha) = (\gamma \cdot (\beta \cdot \alpha)))$ **and**
$C_3 s : \forall \gamma.\ IDMcL(dom\ \gamma) \wedge E(\gamma \cdot (dom\ \gamma))$ **and**
$C_4 s : \forall \gamma.\ IDMcL(cod\ \gamma) \wedge E((cod\ \gamma) \cdot \gamma)$
begin
lemma *True* **nitpick** [*satisfy*] **oops** — Consistency
lemma assumes $\exists x.\ \neg(E\ x)$ **shows** *True* **nitpick** [*satisfy*] **oops** — Consistency
lemma assumes $(\exists x.\ \neg(E\ x)) \wedge (\exists x.\ (E\ x))$ **shows** *True* **nitpick** [*satisfy*] **oops** — Consistency
end

6.4 SkolemizedAxiomsSetMcL entails AxiomsSetMcL and AxiomsSet1-5

context *SkolemizedAxiomsSetMcL*
begin
lemma $C_0 : E(x \cdot y) \rightarrow (E\ x \wedge E\ y)$ **using** $C_0 s$ **by** *blast*
lemma $C_1 : \forall \gamma\ \beta\ \alpha.\ (E(\gamma \cdot \beta) \wedge E((\gamma \cdot \beta) \cdot \alpha)) \rightarrow E(\beta \cdot \alpha)$ **using** $C_1 s$ **by** *blast*
lemma $C_1{}': \forall \gamma\ \beta\ \alpha.\ (E(\beta \cdot \alpha) \wedge E(\gamma \cdot (\beta \cdot \alpha))) \rightarrow E(\gamma \cdot \beta)$ **using** $C_1{}'s$ **by** *blast*
lemma $C_2 : \forall \gamma\ \beta\ \alpha.\ (E(\gamma \cdot \beta) \wedge E(\beta \cdot \alpha)) \rightarrow (E((\gamma \cdot \beta) \cdot \alpha) \wedge E(\gamma \cdot (\beta \cdot \alpha)) \wedge ((\gamma \cdot \beta) \cdot \alpha) = (\gamma \cdot (\beta \cdot \alpha)))$ **using** $C_2 s$ **by** *blast*
lemma $C_3 : \forall \gamma.\ \exists\, eD.\ IDMcL(eD) \wedge E(\gamma \cdot eD)$ **by** (*metis* $C_0 s$ $C_3 s$)
lemma $C_4 : \forall \gamma.\ \exists\, eR.\ IDMcL(eR) \wedge E(eR \cdot \gamma)$ **by** (*metis* $C_0 s$ $C_4 s$)

lemma $S_i: E(x \cdot y) \rightarrow (E\ x \wedge E\ y)$ **using** $C_0 s$ **by** *blast*
lemma $E_i: E(x \cdot y) \leftarrow (E\ x \wedge E\ y \wedge (\exists z.\ z \cdot z \cong z \wedge x \cdot z \cong x \wedge z \cdot y \cong y))$ **by** (*metis* $C_2 s$)
lemma $A_i: x \cdot (y \cdot z) \cong (x \cdot y) \cdot z$ **by** (*metis* $C_1 s$ $C_1{}'s$ $C_2 s$ $C_0 s$)
lemma $C_i: \forall y. \exists i.\ ID\ i \wedge i \cdot y \cong y$ **by** (*metis* $C_0 s$ $C_4 s$)
lemma $D_i: \forall x. \exists j.\ ID\ j \wedge x \cdot j \cong x$ **by** (*metis* $C_0 s$ $C_3 s$)

lemma $S_{ii}: (E(x \cdot y) \rightarrow (E\ x \wedge E\ y)) \wedge (E(dom\ x\ ) \rightarrow E\ x) \wedge (E(cod\ y) \rightarrow E\ y)$ **using** $C_0 s$ **by** *blast*
lemma $E_{ii}: E(x \cdot y) \leftarrow (E\ x \wedge E\ y \wedge (\exists z.\ z \cdot z \cong z \wedge x \cdot z \cong x \wedge z \cdot y \cong y))$ **by** (*metis* $C_2 s$)
lemma $A_{ii}: x \cdot (y \cdot z) \cong (x \cdot y) \cdot z$ **by** (*metis* $C_1 s$ $C_1{}'s$ $C_2 s$ $C_0 s$)
lemma $C_{ii}: E\ y \rightarrow (ID(cod\ y) \wedge (cod\ y) \cdot y \cong y)$ **using** $C_4 s$ **by** *auto*
lemma $D_{ii}: E\ x \rightarrow (ID(dom\ x) \wedge x \cdot (dom\ x) \cong x)$ **using** $C_3 s$ **by** *auto*

— AxiomsSets3/4 are omitted here; we already know they are equivalent.

lemma *S1*: $E(dom\ x) \rightarrow E\ x$ **using** $C_0 s$ **by** *blast*
lemma *S2*: $E(cod\ y) \rightarrow E\ y$ **using** $C_0 s$ **by** *blast*
lemma *S3*: $E(x \cdot y) \leftrightarrow dom\ x \simeq cod\ y$ **by** (*metis* (*full-types*) $C_0 s$ $C_1 s$ $C_1{}'s$ $C_2 s$ $C_3 s$ $C_4 s$)
lemma *S4*: $x \cdot (y \cdot z) \cong (x \cdot y) \cdot z$ **by** (*metis* $C_0 s$ $C_1 s$ $C_1{}'s$ $C_2 s$)
lemma *S5*: $(cod\ y) \cdot y \cong y$ **using** $C_0 s$ $C_4 s$ **by** *blast*
lemma *S6*: $x \cdot (dom\ x) \cong x$ **using** $C_0 s$ $C_3 s$ **by** *blast*
end

References

[1] C. Benzmüller and D. Scott. Automating free logic in Isabelle/HOL. In G.-M. Greuel, T. Koch, P. Paule, and A. Sommese, editors, *Mathematical Software – ICMS 2016, 5th International Congress, Proceedings*, volume 9725 of *LNCS*, pages 43–50, Berlin, Germany, 2016. Springer.

[2] C. Benzmüller and D. Scott. Some reflections on a computer-aided theory exploration study in category theory (extended abstract). In T. C. Hales, C. Kaliszyk, S. Schulz, and J. Urban, editors, *3rd Conference on Artificial Intelligence and Theorem Proving (AITP 2018), Book of Abstracts*, 2018.

[3] C. Benzmüller and D. S. Scott. Axiomatizing category theory in free logic. Technical report, CoRR, 2016. http://arxiv.org/abs/1609.01493.

[4] P. J. Freyd and A. Scedrov. *Categories, Allegories*. North Holland, 1990.

[5] S. McLane. Groups, categories and duality. *Proceedings of the National Academy of Sciences*, 34(6):263–267, 1948.

[6] D. Scott. Identity and existence in intuitionistic logic. In M. Fourman, C. Mulvey, and D. Scott, editors, *Applications of Sheaves: Proceedings of the Research Symposium on Applications of Sheaf Theory to Logic, Algebra, and Analysis, Durham, July 9–21, 1977*, volume 752 of *Lecture Notes in Mathematics*, pages 660–696. Springer Berlin Heidelberg, 1979.